
Introduction

The VUnit iden�ty package ( id_pkg ) enables the crea�on of a hierarchical organiza�on of named objects within a testbench. It expands upon
the func�onality provided by VHDL’s simple_name , instance_name  and path_name  a�ributes and eliminates some of the limita�ons associated
with string-based naming conven�ons.

Limitations of Name Strings and Name Attributes

When naming a testbench object, such as a verifica�on component (VC), we can u�lize one of the VHDL name a�ributes. In the example
below, we have a testbench with two VCs, X and Y, each with a name  generic that we assign the path_name  of the VC en�ty instan�a�on.

architecture vunit_style of tb_dut is
  -- Declarations
begin
  stimuli : process
  begin
    ...
  end process;

  my_dut : entity work.dut
    port map(
      ...
    );

  vc_x : entity work.verification_component_x
    generic map(
      name => vc_x'path_name
    )
    port map(
      ...
    );

  vc_y : entity work.verification_component_y
    generic map(
      name => vc_y'path_name
    )
    port map(
      ...
    );
end architecture;

 Note

A limita�on with using name a�ributes is that, at the �me of wri�ng, not all simulators support assigning the name  generic a value that
references the label of the instan�a�on itself.

The VCs use the name  generic to tag log messages. A basic message, without using VUnit logging func�onality, could appear as follows:

Writing 0xDEADBEEF to address 0x12345678 (:tb_dut:vc_x:).



In this case, the name produced by the path_name  a�ribute, :tb_dut:vc_x: , reflects the logical structure of the testbench and clearly iden�fies
the producer. However, path_name  is limited to code structure and knows nothing about the logical structure of the testbench. These are not
necessarilty the same. For example, if we want to reorganize our testbench by moving some declara�ons in the architecture declara�ve part
to a more local loca�on, we can do this by adding a block statement around the VC instan�a�ons.

local_declarations : block is
  -- Local declarations of signals, constants etc
begin
  vc_x : entity work.verification_component_x
    generic map(
      name => vc_x'path_name
    )
    port map(
      ...
    );

  vc_y : entity work.verification_component_y
    generic map(
      name => vc_y'path_name
    )
    port map(
      ...
    );
end block;

We haven’t changed the logical structure, vc_x  is s�ll part of tb_dut  and should be presented as such. However, the code structure has
been changed and so has the naming:

Writing 0xDEADBEEF to address 0x12345678 (:tb_dut2:local_declarations:vc_x:).

To avoid this problem, as well as the problem of limited simulator support, we could opt to abandon the name a�ributes and instead use
manually cra�ed name strings. Manually cra�ing name strings also provides more flexibility as the name is no longer limited by the rules for
iden�fier naming. For example, we could name the VC :tb_dut:verification component X:  if preferred.

Whether using a name a�ribute or providing the name explicitly, the concept of hierarchy is determined by a naming conven�on and not by
the string  type itself. By using a dedicated type, we can create a more explicit concept and also enable more advanced func�onality.



Identity Basics

To overcome the issues discussed in the preceding sec�on, id_pkg  provides an id_t  type, which is compa�ble with VHDL name a�ributes in
the sense that we can create an iden�ty from such an a�ribute:

variable vc_x_id : id_t;

vc_x_id := get_id(vc_x'path_name);

Or from a string if the logical structure doesn’t match the code structure:

vc_x_id := get_id(":tb_dut:vc_x:");

We can also omit the leading and trailing colons for brevity:

vc_x_id := get_id("tb_dut:vc_x");

The iden�ty returned when calling get_id  represents the last component in the hierarchical path. Calling name (vc_x_id)  will return the name
of that component and full_name(vc_x_id)  returns the full path. However, iden��es are created for each component in the path, and the
parent iden�ty can be obtained by invoking the get_parent  func�on:

print("Name = " & name(vc_x_id));
print("Full name = " & full_name(vc_x_id));

parent_id := get_parent(vc_x_id);
print("Parent name = " & name(parent_id));

Name = vc_x
Full name = tb_dut:vc_x
Parent name = tb_dut



Calling the func�on get_id  only creates iden��es for the components that are missing in the path provided to the func�on. For example,
invoking get_id  with vc_y'path_name  will not create a new iden�ty for tb_dut  since that iden�ty already exists a�er our previous invoka�on:

vc_y_id := get_id(vc_y'path_name); -- Creates an identity for vc_y but not for tb_dut.

Another way to add iden��es is to use get_id  with a parent ID parameter. For example, to create the iden�ty for vc_y , we can use the
following equivalent code:

vc_y_id := get_id("vc_y", parent => parent_id);

To gain a be�er understanding of the iden��es that have been generated by previous get_id  calls, we can u�lize the get_tree  func�on to
view the iden�ty tree with a given iden�ty as its root. For example,

print("This is the tb_dut tree:" & get_tree(parent_id));

will output:

This is the tb_dut tree:
tb_dut
+---vc_x
\---vc_y

The get_tree  func�on returns a string star�ng with a linefeed character (LF) to align the root line of the tree with its other elements. To omit
this ini�al LF character, set the op�onal parameter initial_lf  to false.

We can also call get_tree()  without any parameters to view the full iden�ty tree. This provides a comprehensive overview of all the iden��es
created in user code, by third-party IPs, and in VUnit itself. An example of the output is provided below:



This is the full identity tree:
(root)
+---default
+---vunit_lib
|   +---dictionary
|   \---event_pkg
+---check
+---runner
\---tb_dut
    +---vc_x
    \---vc_y

At the root of the tree is a symbol (root)  which represents the predefined root_id . root_id  has no name but is given a symbol in the tree
representa�on for clarity. The lack of name means that we cannot create a new iden�ty with no name as that is already taken.

We can easily check if an iden�ty is taken by calling the has_id  func�on with either the full name of the iden�ty or a par�al name rela�ve to
its parent iden�ty. For example:

has_id("") = true
has_id("tb_dut:vc_x") = true
has_id("vc_y", parent => get_id("tb_dut")) = true
has_id("vc_x") = false

Structuring Identities

The iden�ty package does not place any limita�ons on what we use iden��es for. However, there are a few recommenda�ons:

1. All iden�fers should have a primary owner, the object that the iden�ty is associated with. For instance, in our prior examples the iden�ty
was associated with a verifica�on component.

2. An iden�ty can be used by objects other than its owner, provided that these objects are ac�ng on the owner’s behalf. For example, a
verifica�on component can create a logger from its iden�ty. The logger logs messages on behalf of the verifica�on component and can
share its iden�ty.

3. Use parent-child rela�onsships when the parent object is composed of child objects. For example, a bus protocol checker can be a
standalone VC that monitors transac�ons on a bus to ensure that they are compliant with the protocol. As a standalone VC, it can have an
iden�ty such as protocol_checker . However, if the protocol checker is built into a bus ini�ator VC, capable of ini�a�ng read and write
transac�ons, then a more descrip�ve iden�ty such as bus_initiator:protocol_checker  is more appropriate.



4. If your object is an en�ty, it should have an iden�ty generic. The en�ty itself cannot determine which func�onal hierarchy it belongs to, so
this must be specified externally. The generic should have null_id  as the default value. If no value is assigned, the en�ty is free to choose
its own iden�ty.

Searching the Identity Tree

The get_tree  func�on collects iden�ty names by traversing the full tree. We can also create our own custom tree-traversing func�onality by
leveraging the get_parent , num_children  and get_child  func�ons. The num_children  func�on returns the number of children iden��es a
given iden�ty possesses and we can retrieve each of these children iden��es by calling get_child  with an index in the range [0, number of
children - 1]. For example:

num_children(get_id("tb_dut"))) = 2
name(get_child(get_id("tb_dut"), 1)) = vc_y

To further illustrate these func�ons we can examine how vc_x  handles the situa�on when its id  generic hasn’t been assinged any value and
defaults to null_id . Rather than simply taking a fix iden�ty name which would be shared by all instances, or an instance_name  which may or
may not yield a good representa�on, it creates another logical structure based on the format <company name>:<VC name>:<instance
number>. The instance number is calculated by searching the company iden�ty space for other exis�ng instances of vc_x . If n instances are
found, the new instance is assigned number n + 1.



if id = null_id then
  acme_corp_id := get_id("Acme Corporation");
  num_vc_x := 0;
  for child_idx in 0 to num_children(acme_corp_id) - 1 loop
    if name(get_child(acme_corp_id, child_idx)) = "vc_x" then
      num_vc_x := num_vc_x + 1;
    end if;
  end loop;
  vc_x_id := get_id("vc_x", parent => acme_corp_id);
  my_id := get_id(to_string(num_vc_x + 1), parent => vc_x_id);
  logger := get_logger(my_id);
else
  logger := get_logger(id);
end if;

...

debug(logger, "Writing 0x" & to_hstring(data) & " to address 0x" & to_hstring(addr) & ".");

With only one vc_x , the instance will be designated as number 1.

10000000 fs - Acme Corporation:vc_x:1 -   DEBUG - Writing 0xDEADBEEF to address 0x12345678.

In this example we were able to search for other instances of vc_x  by searching the iden�ty namespace. Had there not been a naming
conven�on for the vc_x  instances, it would not have been possible. However, not all resources have such a conven�on. For instance, if a VC
wants to use the closest exis�ng logger among its ancestors, there is no naming conven�on to rely on. In such cases, get_logger  is of no use
as this procedure will only create a new logger if it doesn’t already exist. Fortunately, there is another func�on, has_logger , which can be used
to query if there is a logger for an exis�ng iden�ty. All in all, it is generally recommended that any resource which u�lizes iden��es should
also provide methods for determining the existence of such a resource for a given iden�ty.


